/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

dependencies {
    // https://docs.gradle.org/8.3/userguide/upgrading_version_8.html#test_framework_implementation_dependencies
    add('testRuntimeOnly', 'org.junit.platform:junit-platform-launcher')
}

tasks.named('compileTestGroovy', GroovyCompile) {
    groovyOptions.forkOptions.jvmArgs = ['-Xmx768m']
}

tasks.withType(Test).configureEach {
    onlyIf {
        ![
                'onlyFunctionalTests',
                'onlyHibernate5Tests',
                'skipMongodbTests',
                'onlyCoreTests',
                'skipTests'
        ].find {
            project.hasProperty(it)
        }
    }

    useJUnitPlatform()
    maxParallelForks = 1
    jvmArgs = ['-Xmx1028M']
    afterSuite {
        System.out.print('.')
        System.out.flush()
    }
    testLogging {
        events('passed', 'skipped', 'failed')
        exceptionFormat = 'full'
        showCauses = true
        showExceptions = true
        showStackTraces = true
    }
    // Used in the TCK test to selectively enable/disable tests
    systemProperty('mongodb.gorm.suite', 'true')
}

